Exploring Simplified Variants of Gödel's Ontological Argument in Isabelle/HOL

Christoph Benzmüller

December 14, 2021

Abstract

Simplified variants of Gödel's ontological argument are explored. Among those is a particularly interesting simplified argument which is (i) valid already in basic modal logics K or KT, (ii) which does not suffer from modal collapse, and (iii) which avoids the rather complex predicates of essence (Ess.) and necessary existence (NE) as used by Gödel.

Whether the presented variants increase or decrease the attractiveness and persuasiveness of the ontological argument is a question I would like to pass on to philosophy and theology.

Contents

1 Background Reading

The selected simplified variants of Gödel's ontological argument [5, 7] as presented in §3 have first been extracted from the insights gained in §6 of [4]. These variants are also influenced by the work presented in [1] and they significantly extend the findings from [3]. In §4 we additionally include the sources from [4].

2 Higher-Order Modal Logic in HOL (cf. [2] and Fig. 1 in [4]).

theory *HOML* **imports** *Main*
begin
nitpick-params[*user-axioms*,*expect*=*genuine*]

Type i is associated with possible worlds and type e with entities:

typedecl i — Possible worlds
typedecl e — Individuals
type-synonym $\sigma = i{\Rightarrow}bool$ — World-lifted propositions
type-synonym $\gamma = e{\Rightarrow}\sigma$ — Lifted predicates
type-synonym $\mu = \sigma{\Rightarrow}\sigma$ — Unary modal connectives
type-synonym $\nu = \sigma{\Rightarrow}\sigma{\Rightarrow}\sigma$ — Binary modal connectives

Logical connectives (operating on truth-sets):

abbreviation $c1{::}\sigma$ ($\bot$) **where** $\bot \equiv \lambda w.\ False$
abbreviation $c2{::}\sigma$ ($\top$) **where** $\top \equiv \lambda w.\ True$
abbreviation $c3{::}\mu$ ($\neg$-[52]53) **where** $\neg\varphi \equiv \lambda w.\neg(\varphi\ w)$
abbreviation $c4{::}\nu$ (**infix**$\wedge$50) **where** $\varphi\wedge\psi \equiv \lambda w.(\varphi\ w)\wedge(\psi\ w)$
abbreviation $c5{::}\nu$ (**infix**$\vee$49) **where** $\varphi\vee\psi \equiv \lambda w.(\varphi\ w)\vee(\psi\ w)$
abbreviation $c6{::}\nu$ (**infix**$\rightarrow$48) **where** $\varphi\rightarrow\psi \equiv \lambda w.(\varphi\ w)\longrightarrow(\psi\ w)$
abbreviation $c7{::}\nu$ (**infix**$\leftrightarrow$47) **where** $\varphi\leftrightarrow\psi \equiv \lambda w.(\varphi\ w)\longleftrightarrow(\psi\ w)$
consts $R{::}i{\Rightarrow}i{\Rightarrow}bool$ (-**r**-) — Accessibility relation
abbreviation $c8{::}\mu$ ($\Box$-[54]55) **where** $\Box\varphi \equiv \lambda w.\forall v.(w\mathbf{r}v)\longrightarrow(\varphi\ v)$
abbreviation $c9{::}\mu$ ($\Diamond$-[54]55) **where** $\Diamond\varphi \equiv \lambda w.\exists v.(w\mathbf{r}v)\wedge(\varphi\ v)$
abbreviation $c10{::}\gamma{\Rightarrow}\gamma$ ($\neg$-[52]53) **where** $\neg\Phi \equiv \lambda x.\lambda w.\neg(\Phi\ x\ w)$
abbreviation $c11{::}\gamma{\Rightarrow}\gamma$ ($\rightharpoondown$-) **where** $\rightharpoondown\Phi \equiv \lambda x.\lambda w.\neg(\Phi\ x\ w)$
abbreviation $c12{::}e{\Rightarrow}e{\Rightarrow}\sigma$ (-=-) **where** $x{=}y \equiv \lambda w.(x{=}y)$
abbreviation $c13{::}e{\Rightarrow}e{\Rightarrow}\sigma$ (-$\neq$-) **where** $x{\neq}y \equiv \lambda w.(x{\neq}y)$

Polymorphic possibilist quantification:

abbreviation *q1*::$('a \Rightarrow \sigma) \Rightarrow \sigma$ ($\forall$) **where** $\forall\,\Phi \equiv \lambda w. \forall\, x.(\Phi\ x\ w)$
abbreviation *q2* (**binder**$\forall$ [*10*]*11*) **where** $\forall\, x.\ \varphi(x) \equiv \forall\, \varphi$
abbreviation *q3*::$('a \Rightarrow \sigma) \Rightarrow \sigma$ ($\exists$) **where** $\exists\,\Phi \equiv \lambda w. \exists\, x.(\Phi\ x\ w)$
abbreviation *q4* (**binder**$\exists$ [*10*]*11*) **where** $\exists\, x.\ \varphi(x) \equiv \exists\, \varphi$

Actualist quantification for individuals/entities:

consts *existsAt*::γ (-@-)
abbreviation *q5*::$\gamma \Rightarrow \sigma$ ($\forall^E$) **where** $\forall^E \Phi \equiv \lambda w. \forall\, x.(x@w) \longrightarrow (\Phi\ x\ w)$
abbreviation *q6* (**binder**$\forall^E$[*8*]*9*) **where** $\forall^E x.\ \varphi(x) \equiv \forall^E \varphi$
abbreviation *q7*::$\gamma \Rightarrow \sigma$ ($\exists^E$) **where** $\exists^E \Phi \equiv \lambda w. \exists\, x.(x@w) \wedge (\Phi\ x\ w)$
abbreviation *q8* (**binder**$\exists^E$[*8*]*9*) **where** $\exists^E x.\ \varphi(x) \equiv \exists^E \varphi$

Meta-logical predicate for global validity:

abbreviation *g1*::$\sigma \Rightarrow$*bool* ($\lfloor$-$\rfloor$) **where** $\lfloor\psi\rfloor \equiv\ \forall\, w.\ \psi\ w$

Barcan and converse Barcan formula:

lemma *True* **nitpick**[*satisfy*] **oops** — Model found by Nitpick
lemma $\lfloor(\forall^E x.\Box(\varphi\ x)) \rightarrow \Box(\forall^E x.(\varphi\ x))\rfloor$ **nitpick oops** — Ctm
lemma $\lfloor\Box(\forall^E x.(\varphi\ x)) \rightarrow (\forall^E x.\Box(\varphi\ x))\rfloor$ **nitpick oops** — Ctm
lemma $\lfloor(\forall\, x.\Box(\varphi\ x)) \rightarrow \Box(\forall\, x.\ \varphi\ x)\rfloor$ **by** *simp*
lemma $\lfloor\Box(\forall\, x.(\varphi\ x)) \rightarrow (\forall\, x.\Box(\varphi\ x))\rfloor$ **by** *simp*
end
theory *DisableKodkodScala*
imports *Main*
begin

Some of the nitpick invocation within this AFP entry do not work, if "Kodkod Scala" is enabled, i.e., if the box under Plugin Options — Isabelle — General — Miscelleaneous Tools — Kodkod Scala is activated. Therefore, in this theory we explicitly disable this configuration option.

ML ‹
Options.default-put-bool ***system-option*** ‹*kodkod-scala*› *false*
›

end

3 Selected Simplified Ontological Argument

theory *SimplifiedOntologicalArgument* **imports**
HOML
DisableKodkodScala
begin

Positive properties:

consts *posProp*::$\gamma \Rightarrow \sigma$ ($\mathcal{P}$)

An entity x is God-like if it possesses all positive properties.

definition *G* ($\mathcal{G}$) **where** $\mathcal{G}(x) \equiv \forall\,\Phi.(\mathcal{P}(\Phi) \rightarrow \Phi(x))$

The axiom's of the simplified variant are presented next; these axioms are further motivated in [4, 1]).

Self-difference is not a positive property (possible alternative: the empty property is not a positive property).

axiomatization where *CORO1*: $\lfloor \neg(\mathcal{P}(\lambda x.(x \neq x))) \rfloor$

A property entailed by a positive property is positive.

axiomatization where *CORO2*: $\lfloor \forall\, \Phi\ \Psi.\ \mathcal{P}(\Phi) \wedge (\forall\, x.\ \Phi(x) \rightarrow \Psi(x)) \rightarrow \mathcal{P}(\Psi) \rfloor$

Being Godlike is a positive property.

axiomatization where *AXIOM3*: $\lfloor \mathcal{P}\ \mathcal{G} \rfloor$

3.1 Verifying the Selected Simplified Ontological Argument (version 1)

The existence of a non-exemplified positive property implies that self-difference (or, alternatively, the empty property) is a positive property.

lemma *LEMMA1*: $\lfloor (\exists\, \Phi.(\mathcal{P}(\Phi) \wedge \neg(\exists\, x.\ \Phi(x)))) \rightarrow \mathcal{P}(\lambda x.(x \neq x)) \rfloor$
  using *CORO2* **by** *meson*

A non-exemplified positive property does not exist.

lemma *LEMMA2*: $\lfloor \neg(\exists\, \Phi.(\mathcal{P}(\Phi) \wedge \neg(\exists\, x.\ \Phi(x)))) \rfloor$
  using *CORO1 LEMMA1* **by** *blast*

Positive properties are exemplified.

lemma *LEMMA3*: $\lfloor \forall\, \Phi.(\mathcal{P}(\Phi) \rightarrow (\exists\, x.\ \Phi(x))) \rfloor$
  using *LEMMA2* **by** *blast*

There exists a God-like entity.

theorem *THEOREM3'*: $\lfloor \exists\, x.\ \mathcal{G}(x) \rfloor$
  using *AXIOM3 LEMMA3* **by** *auto*

Necessarily, there exists a God-like entity.

theorem *THEOREM3*: $\lfloor \Box(\exists\, x.\ \mathcal{G}(x)) \rfloor$
  using *THEOREM3'* **by** *simp*

However, the possible existence of Godlike entity is not implied.

theorem *CORO*: $\lfloor \Diamond(\exists\, x.\ \mathcal{G}(x)) \rfloor$
  nitpick oops

3.2 Verifying the Selected Simplified Ontological Argument (version 2)

We switch to logic T.

axiomatization where *T*: $\lfloor \forall\, \varphi.\ \Box\varphi \rightarrow \varphi \rfloor$
lemma *T'*: $\lfloor \forall\, \varphi.\ \varphi \rightarrow \Diamond\varphi \rfloor$ **using** *T* **by** *metis*

Positive properties are possibly exemplified.

theorem *THEOREM1*: $\lfloor\forall\,\Phi.\ \mathcal{P}(\Phi) \rightarrow \Diamond(\exists\,x.\ \Phi(x))\rfloor$
 using *CORO1 CORO2 T′* **by** *metis*

Possibly there exists a God-like entity.

theorem *CORO*: $\lfloor\Diamond(\exists\,x.\ \mathcal{G}(x))\rfloor$
 using *AXIOM3 THEOREM1* **by** *auto*

The possible existence of a God-like entity impplies the necessary existence of a God-like entity.

theorem *THEOREM2*: $\lfloor\Diamond(\exists\,x.\ \mathcal{G}(x)) \rightarrow \Box(\exists\,x.\ \mathcal{G}(x))\rfloor$
 using *AXIOM3 CORO1 CORO2* **by** *metis*

Necessarily, there exists a God-like entity.

theorem *THEO3*: $\lfloor\Box(\exists\,x.\ \mathcal{G}(x))\rfloor$
 using *CORO THEOREM2* **by** *blast*

There exists a God-like entity.

theorem *THEO3′*: $\lfloor\exists\,x.\ \mathcal{G}(x)\rfloor$
 using *T THEO3* **by** *metis*

Modal collapse is not implied; nitpick reports a countermodel.

lemma *MC*: $\lfloor\forall\,\Phi.\ \Phi \rightarrow \Box\Phi\rfloor$ **nitpick oops**

Consistency of the theory; nitpick reports a model.

lemma *True* **nitpick**[*satisfy*] **oops**
end

4 Presentation of All Variants as Studied in [4]

4.1 Preliminaries: Modal Ultrafilter (Fig. 2 in [4])

theory *MFilter* **imports** *HOML*
begin

Some abbreviations for auxiliary operations.

abbreviation $a{::}\gamma\Rightarrow(\gamma\Rightarrow\sigma)\Rightarrow\sigma$ (-∈-) **where** $x{\in}S \equiv S\ x$
abbreviation $b{::}\gamma$ ($\emptyset$) **where** $\emptyset \equiv \lambda x.\ \bot$
abbreviation $c{::}\gamma$ ($\mathbf{U}$) **where** $\mathbf{U} \equiv \lambda x.\ \top$
abbreviation $d{::}\gamma\Rightarrow\gamma\Rightarrow\sigma$ (-⊆-) **where** $\varphi{\subseteq}\psi \equiv \forall\,x.((\varphi\ x) \rightarrow (\psi\ x))$
abbreviation $e{::}\gamma\Rightarrow\gamma\Rightarrow\gamma$ (-⊓-) **where** $\varphi{\sqcap}\psi \equiv \lambda x.((\varphi\ x) \wedge (\psi\ x))$
abbreviation $f{::}\gamma\Rightarrow\gamma$ ($^{-1}$-) **where** $^{-1}\psi \equiv \lambda x.\ \neg(\psi\ x)$

Definition of modal filter.

abbreviation $g{::}(\gamma\Rightarrow\sigma)\Rightarrow\sigma$ (*Filter*)
 where *Filter* $\Phi \equiv (((\mathbf{U}{\in}\Phi) \wedge \neg(\emptyset{\in}\Phi))$
 $\wedge\ (\forall\,\varphi\ \psi.\ (((\varphi{\in}\Phi) \wedge (\varphi{\subseteq}\psi)) \rightarrow (\psi{\in}\Phi))))$
 $\wedge\ (\forall\,\varphi\ \psi.\ (((\varphi{\in}\Phi) \wedge (\psi{\in}\Phi)) \rightarrow ((\varphi{\sqcap}\psi){\in}\Phi)))$

Definition of modal ultrafilter .

abbreviation h::($\gamma\Rightarrow\sigma$)$\Rightarrow\sigma$ (*UFilter*) **where**
UFilter $\Phi \equiv$ (*Filter* Φ)$\wedge$($\forall\,\varphi$.(($\varphi\in\Phi$) $\vee$ (($^{-1}\varphi$)$\in\Phi$)))

Modal filter and modal ultrafilter are consistent.

lemma $\lfloor\forall\,\Phi\ \varphi.(($*UFilter* $\Phi) \rightarrow \neg((\varphi\in\Phi) \wedge ((^{-1}\varphi)\in\Phi)))\rfloor$ **by** *force*
end

4.2 Preliminaries: Further Basic Notions (Fig. 3 in [4])

theory *BaseDefs* **imports** *HOML*
begin

Positive properties.

consts *posProp*::$\gamma\Rightarrow\sigma$ ($\mathcal{P}$)

Basic definitions for modal ontological argument.

abbreviation a (-$\sqsubseteq$-) **where** $X\sqsubseteq Y \equiv \forall^E z.((X\ z) \rightarrow (Y\ z))$
abbreviation b (-$\Rrightarrow$-) **where** $X\Rrightarrow Y \equiv \Box(X\sqsubseteq Y)$
abbreviation c ($\mathcal{P}$os) **where** $\mathcal{P}$os $Z \equiv \forall\,X.((Z\ X) \rightarrow (\mathcal{P}\ X))$
abbreviation d (-$\prod$-) **where** $X\prod\mathcal{Z} \equiv \Box(\forall^E u.((X\ u) \leftrightarrow (\forall\,Y.((\mathcal{Z}\ \ Y) \rightarrow (Y\ u)))))$

Definition of Godlike.

definition G ($\mathcal{G}$) **where** $\mathcal{G}\ x \equiv \forall\,Y.((\mathcal{P}\ \ Y) \rightarrow (Y\ x))$

Definitions of Essence and Necessary Existence.

definition E ($\mathcal{E}$) **where** $\mathcal{E}\ \ Y\ x \equiv (Y\ x) \wedge (\forall\,Z.((Z\ x) \rightarrow (Y\Rrightarrow Z)))$
definition *NE* ($\mathcal{NE}$) **where** $\mathcal{NE}\ \ x \equiv \forall\,Y.((\mathcal{E}\ \ Y\ x) \rightarrow \Box(\exists^E\ \ Y))$
end

4.3 Ultrafilter Analysis of Scott's Variant (Fig. 3 in [4]))

theory *ScottVariant* **imports**
HOML
MFilter
BaseDefs
begin

Axioms of Scott's variant.

axiomatization where
A1: $\lfloor\forall\,X.((\neg(\mathcal{P}\ \ X)) \leftrightarrow (\mathcal{P}(\rightharpoondown X)))\rfloor$ **and**
A2: $\lfloor\forall\,X\ \,Y.(((\mathcal{P}\ \ X) \wedge (X\Rrightarrow Y)) \rightarrow (\mathcal{P}\ \ Y))\rfloor$ **and**
A3: $\lfloor\forall\,\mathcal{Z}.((\mathcal{P}\text{os}\ \mathcal{Z}) \rightarrow (\forall\,X.((X\prod\mathcal{Z}) \rightarrow (\mathcal{P}\ \ X))))\rfloor$ **and**
A4: $\lfloor\forall\,X.((\mathcal{P}\ \ X) \rightarrow \Box(\mathcal{P}\ \ X))\rfloor$ **and**
A5: $\lfloor\mathcal{P}\ \mathcal{NE}\rfloor$ **and**
B: $\lfloor\forall\,\varphi.(\varphi \rightarrow \Box\Diamond\varphi)\rfloor$ — Logic KB

lemma B': $\forall\,x\ y.\ \neg(x\mathbf{r}y) \vee (y\mathbf{r}x)$ **using** B **by** *fastforce*

Necessary existence of a Godlike entity.

theorem *T6*: $\lfloor \Box(\exists^E \mathcal{G}) \rfloor$
proof –
have *T1*: $\lfloor \forall X.((\mathcal{P}\ X) \rightarrow \Diamond(\exists^E X)) \rfloor$ **using** *A1 A2* **by** *blast*
have *T2*: $\lfloor \mathcal{P}\ \mathcal{G} \rfloor$ **by** (*metis A3 G-def*)
have *T3*: $\lfloor \Diamond(\exists^E \mathcal{G}) \rfloor$ **using** *T1 T2* **by** *simp*
have *T4*: $\lfloor \forall^E x.((\mathcal{G}\ x) \rightarrow (\mathcal{E}\ \mathcal{G}\ x)) \rfloor$ **unfolding** *G-def E-def* **using** *A1 A4* **by** *metis*
have *T5*: $\lfloor (\Diamond(\exists^E \mathcal{G})) \rightarrow \Box(\exists^E \mathcal{G}) \rfloor$ **by** (*smt A5 G-def B′ NE-def T4*)
thus *?thesis* **using** *T3* **by** *blast* **qed**

Existence of a Godlike entity.

lemma $\lfloor \exists^E \mathcal{G} \rfloor$ **using** *A1 A2 B′ T6* **by** *blast*

Consistency

lemma *True* **nitpick**[*satisfy*] **oops** — Model found.

Modal collapse: holds.

lemma *MC*: $\lfloor \forall \Phi.(\Phi \rightarrow \Box\Phi) \rfloor$
proof – **{**
fix *w* **fix** *Q*
have *1*: $\forall x.((\mathcal{G}\ x\ w) \longrightarrow (\forall Z.((Z\ x) \rightarrow \Box(\forall^E z.((\mathcal{G}\ z) \rightarrow (Z\ z))))) w)$
by (*metis A1 A4 G-def*)
have *2*: $(\exists x.\ \mathcal{G}\ x\ w) \longrightarrow ((Q \rightarrow \Box(\forall^E z.((\mathcal{G}\ z) \rightarrow Q)))\ w)$
using *1* **by** *force*
have *3*: $(Q \rightarrow \Box Q)\ w$ **using** *B′ T6 2* **by** *blast***}**
thus *?thesis* **by** *auto* **qed**

Analysis of positive properties using ultrafilters.

theorem *U1*: $\lfloor UFilter\ \mathcal{P} \rfloor$ — Proof found by sledgehammer
proof –
have *1*: $\lfloor (\mathbf{U} \in \mathcal{P}) \wedge \neg(\emptyset \in \mathcal{P}) \rfloor$ **using** *A1 A2* **by** *blast*
have *2*: $\lfloor \forall \varphi\ \psi.(((\varphi \in \mathcal{P}) \wedge (\varphi \sqsubseteq \psi)) \rightarrow (\psi \in \mathcal{P})) \rfloor$ **by** (*smt A2 B′ MC*)
have *3*: $\lfloor \forall \varphi\ \psi.(((\varphi \in \mathcal{P}) \wedge (\psi \in \mathcal{P})) \rightarrow ((\varphi \sqcap \psi) \in \mathcal{P})) \rfloor$ **by** (*metis A1 A2 G-def B′ T6*)
have *4*: $\lfloor \forall \varphi.((\varphi \in \mathcal{P}) \vee ((^{-1}\varphi) \in \mathcal{P})) \rfloor$ **using** *A1* **by** *blast*
thus *?thesis* **using** *1 2 3 4* **by** *simp* **qed**

lemma *L1*: $\lfloor \forall X\ Y.((X \Rrightarrow Y) \rightarrow (X \sqsubseteq Y)) \rfloor$ **by** (*metis A1 A2 MC*)
lemma *L2*: $\lfloor \forall X\ Y.(((\mathcal{P}\ X) \wedge (X \sqsubseteq Y)) \rightarrow (\mathcal{P}\ Y)) \rfloor$ **by** (*smt A2 B′ MC*)

Set of supersets of X, we call this HF X.

abbreviation *HF* **where** $HF\ X \equiv \lambda Y.(X \sqsubseteq Y)$

$HF\ \mathcal{G}$ is a filter; hence, $HF\ \mathcal{G}$ is Hauptfilter of $\mathcal{G}$.

lemma *F1*: $\lfloor Filter\ (HF\ \mathcal{G}) \rfloor$ **by** (*metis A2 B′ T6 U1*)
lemma *F2*: $\lfloor UFilter\ (HF\ \mathcal{G}) \rfloor$ **by** (*smt A1 F1 G-def*)

T6 follows directly from F1.

theorem *T6again*: $\lfloor \Box(\exists^E \mathcal{G}) \rfloor$ **using** *F1* **by** *simp*
end

4.4 Ultrafilter Variant (Fig. 5 in [4])

theory *UFilterVariant* **imports**
HOML
MFilter
BaseDefs
DisableKodkodScala
begin

Axiom's of ultrafilter variant.

axiomatization where
U1: $\lfloor \mathit{UFilter}\ \mathcal{P} \rfloor$ **and**
A2: $\lfloor \forall X\ Y.(((\mathcal{P}\ X) \wedge (X \Rrightarrow Y)) \rightarrow (\mathcal{P}\ Y)) \rfloor$ **and**
A3: $\lfloor \forall \mathcal{Z}.((\mathcal{P}\mathrm{os}\ \mathcal{Z}) \rightarrow (\forall X.((X \sqcap \mathcal{Z}) \rightarrow (\mathcal{P}\ X)))) \rfloor$

Necessary existence of a Godlike entity.

theorem *T6*: $\lfloor \Box(\exists^E\ \mathcal{G}) \rfloor$ — Proof also found by sledgehammer
proof –
have *T1*: $\lfloor \forall X.((\mathcal{P}\ X) \rightarrow \Diamond(\exists^E\ X)) \rfloor$ **by** (*metis A2 U1*)
have *T2*: $\lfloor \mathcal{P}\ \mathcal{G} \rfloor$ **by** (*metis A3 G-def*)
have *T3*: $\lfloor \Diamond(\exists^E\ \mathcal{G}) \rfloor$ **using** *T1 T2* **by** *simp*
have *T5*: $\lfloor (\Diamond(\exists^E\ \mathcal{G})) \rightarrow \Box(\exists^E\ \mathcal{G}) \rfloor$ **by** (*metis A2 G-def T2 U1*)
thus *?thesis* **using** *T3* **by** *blast* **qed**

Checking for consistency.

lemma *True* **nitpick**[*satisfy*] **oops** — Model found

Checking for modal collapse.

lemma *MC*: $\lfloor \forall \Phi.(\Phi \rightarrow \Box\Phi) \rfloor$ **nitpick oops** — Countermodel
end

4.5 Simplified Variant (Fig. 6 in [4])

theory *SimpleVariant* **imports**
HOML
MFilter
BaseDefs
DisableKodkodScala
begin

Axiom's of new, simplified variant.

axiomatization where
A1': $\lfloor \neg(\mathcal{P}(\lambda x.(x \neq x))) \rfloor$ **and**
A2': $\lfloor \forall X\ Y.(((\mathcal{P}\ X) \wedge ((X \sqsubseteq Y) \vee (X \Rrightarrow Y))) \rightarrow (\mathcal{P}\ Y)) \rfloor$ **and**
A3: $\lfloor \forall \mathcal{Z}.((\mathcal{P}\mathrm{os}\ \mathcal{Z}) \rightarrow (\forall X.((X \sqcap \mathcal{Z}) \rightarrow (\mathcal{P}\ X)))) \rfloor$

lemma *T2*: $\lfloor \mathcal{P}\ \mathcal{G} \rfloor$ **by** (*metis A3 G-def*) — From A3
lemma *L1*: $\lfloor \mathcal{P}(\lambda x.(x{=}x)) \rfloor$ **by** (*metis A2′ A3*)

Necessary existence of a Godlike entity.

theorem *T6*: $\lfloor \Box(\exists^E\ \mathcal{G}) \rfloor$ — Proof found by sledgehammer
proof –
have *T1*: $\lfloor \forall X.((\mathcal{P}\ X) \rightarrow \Diamond(\exists^E\ X)) \rfloor$ **by** (*metis A1′ A2′*)
have *T3*: $\lfloor \Diamond(\exists^E\ \mathcal{G}) \rfloor$ **using** *T1 T2* **by** *simp*
have *T5*: $\lfloor (\Diamond(\exists^E\ \mathcal{G})) \rightarrow \Box(\exists^E\ \mathcal{G}) \rfloor$ **by** (*metis A1′ A2′ T2*)
thus *?thesis* **using** *T3* **by** *blast* **qed**

lemma *True* **nitpick**[*satisfy*] **oops** — Consistency: model found

Modal collapse and monotheism: not implied.

lemma *MC*: $\lfloor \forall \Phi.(\Phi \rightarrow \Box\Phi) \rfloor$ **nitpick oops** — Countermodel
lemma *MT*: $\lfloor \forall x\ y.(((\mathcal{G}\ x) \wedge (\mathcal{G}\ y)) \rightarrow (x{=}y)) \rfloor$
nitpick oops — Countermodel.

Gödel's A1, A4, A5: not implied anymore.

lemma *A1*: $\lfloor \forall X.((\neg(\mathcal{P}\ X)) \leftrightarrow (\mathcal{P}(\rightharpoondown X))) \rfloor$ **nitpick oops** — Countermodel
lemma *A4*: $\lfloor \forall X.((\mathcal{P}\ X) \rightarrow \Box(\mathcal{P}\ X)) \rfloor$ **nitpick oops** — Countermodel
lemma *A5*: $\lfloor \mathcal{P}\ \mathcal{NE} \rfloor$ **nitpick oops** — Countermodel

Checking filter and ultrafilter properties.

theorem *F1*: $\lfloor$*Filter* $\mathcal{P}\rfloor$ **oops** — Proof found by sledgehammer, but reconstruction timeout
theorem *U1*: $\lfloor$*UFilter* $\mathcal{P}\rfloor$ **nitpick oops** — Countermodel
end

4.6 Simplified Variant with Axiom T2 (Fig. 7 in [4])

theory *SimpleVariantPG* **imports**
HOML
MFilter
BaseDefs
DisableKodkodScala
begin

Axiom's of simplified variant with A3 replaced.

axiomatization where
A1′: $\lfloor \neg(\mathcal{P}(\lambda x.(x{\neq}x))) \rfloor$ **and**
A2′: $\lfloor \forall X\ Y.(((\mathcal{P}\ X) \wedge ((X \sqsubseteq Y) \vee (X \Rrightarrow Y))) \rightarrow (\mathcal{P}\ Y)) \rfloor$ **and**
T2: $\lfloor \mathcal{P}\ \mathcal{G} \rfloor$

Necessary existence of a Godlike entity.

theorem *T6*: $\lfloor \Box(\exists^E\ \mathcal{G}) \rfloor$ — Proof found by sledgehammer
proof –
have *T1*: $\lfloor \forall X.((\mathcal{P}\ X) \rightarrow \Diamond(\exists^E\ X)) \rfloor$ **by** (*metis A1′ A2′*)
have *T3*: $\lfloor \Diamond(\exists^E\ \mathcal{G}) \rfloor$ **using** *T1 T2* **by** *simp*

have *T5*: $\lfloor(\Diamond(\exists^E \mathcal{G})) \rightarrow \Box(\exists^E \mathcal{G})\rfloor$ **by** (*metis A1′ A2′ T2*)
thus *?thesis* **using** *T3* **by** *blast* **qed**

lemma *True* **nitpick**[*satisfy*] **oops** — Consistency: model found

Modal collapse and Monotheism: not implied.

lemma *MC*: $\lfloor\forall \Phi.(\Phi \rightarrow \Box\Phi)\rfloor$ **nitpick oops** — Countermodel
lemma *MT*: $\lfloor\forall x\, y.(((\mathcal{G}\ x)\wedge(\mathcal{G}\ y))\rightarrow(x=y))\rfloor$ **nitpick oops** — Countermodel
end

4.7 Simplified Variant with Simple Entailment in Logic K (Fig. 8 in [4])

theory *SimpleVariantSE* **imports**
HOML
MFilter
BaseDefs
begin

Axiom's of new variant based on ultrafilters.

axiomatization where
A1′: $\lfloor\neg(\mathcal{P}(\lambda x.(x\neq x)))\rfloor$ **and**
A2″: $\lfloor\forall X\ Y.(((\mathcal{P}\ X) \wedge (X \sqsubseteq Y)) \rightarrow (\mathcal{P}\ Y))\rfloor$ **and**
T2: $\lfloor\mathcal{P}\ \mathcal{G}\rfloor$

Necessary existence of a Godlike entity.

theorem *T6*: $\lfloor\Box(\exists^E \mathcal{G})\rfloor$ **using** *A1′ A2″ T2* **by** *blast*
theorem *T7*: $\lfloor\exists^E \mathcal{G}\rfloor$ **using** *A1′ A2″ T2* **by** *blast*

Possible existence of a Godlike: has counterodel.

lemma *T3*: $\lfloor\Diamond(\exists^E \mathcal{G})\rfloor$ **nitpick oops** — Countermodel

lemma *T3′*: **assumes** *T*: $\lfloor\forall \varphi.((\Box\varphi) \rightarrow \varphi)\rfloor$
shows $\lfloor\Diamond(\exists^E \mathcal{G})\rfloor$
using *A1′ A2″ T2 T* **by** *metis*
end

4.8 Simplified Variant with Simple Entailment in Logic T (Fig. 9 in [4])

theory *SimpleVariantSEinT* **imports**
HOML
MFilter
BaseDefs
begin

Axiom's of new variant based on ultrafilters.

axiomatization where
A1′: $\lfloor\neg(\mathcal{P}(\lambda x.(x\neq x)))\rfloor$ **and**

A2'': $\lfloor\forall X\ Y.(((\mathcal{P}\ X) \wedge (X\sqsubseteq Y)) \rightarrow (\mathcal{P}\ Y))\rfloor$ **and**
T2: $\lfloor\mathcal{P}\ \mathcal{G}\rfloor$

Modal Logic T.

axiomatization where *T*: $\lfloor\forall \varphi.((\Box\varphi) \rightarrow \varphi)\rfloor$
lemma *T'*: $\lfloor\forall \varphi.(\varphi \rightarrow (\Diamond\varphi))\rfloor$ **by** (*metis T*)

Necessary existence of a Godlike entity.

theorem *T6*: $\lfloor\Box(\exists^E\ \mathcal{G})\rfloor$ — Proof found by sledgehammer
proof –
have *T1*: $\lfloor\forall X.((\mathcal{P}\ X)\rightarrow(\Diamond(\exists^E\ X)))\rfloor$ **by** (*metis A1' A2'' T'*)
have *T3*: $\lfloor\Diamond(\exists^E\ \mathcal{G})\rfloor$ **by** (*metis T1 T2*)
have *T5*: $\lfloor(\Diamond(\exists^E\ \mathcal{G})) \rightarrow \Box(\exists^E\ \mathcal{G})\rfloor$ **by** (*metis A1' A2'' T2*)
thus *?thesis* **using** *T3* **by** *simp* **qed**

T6 again, with an alternative, simpler proof.

theorem *T6again*: $\lfloor\Box(\exists^E\ \mathcal{G})\rfloor$
proof –
have *L1*: $\lfloor(\exists X.((\mathcal{P}\ X)\wedge\neg(\exists^E X)))\rightarrow(\mathcal{P}(\lambda x.(x\neq x)))\rfloor$
by (*smt A2''*)
have *L2*: $\lfloor\neg(\exists X.((\mathcal{P}\ X) \wedge \neg(\exists^E\ X)))\rfloor$ **by** (*metis L1 A1'*)
have *T1'*: $\lfloor\forall X.((\mathcal{P}\ X) \rightarrow (\exists^E\ X))\rfloor$ **by** (*metis L2*)
have *T3'*: $\lfloor\exists^E\ \mathcal{G}\rfloor$ **by** (*metis T1' T2*)
have *L3*: $\lfloor\Diamond(\exists^E\ \mathcal{G})\rfloor$ **by** (*metis T3' T'*) — not needed
thus *?thesis* **using** *T3'* **by** *simp* **qed**
end

4.9 Hauptfiltervariant (Fig. 10 in [4])

theory *SimpleVariantHF* **imports**
HOML
MFilter
BaseDefs
DisableKodkodScala
begin

Definition: Set of supersets of X, we call this $\mathcal{HF}\ X$.

abbreviation *HF*::$\gamma\Rightarrow(\gamma\Rightarrow\sigma)$ **where** *HF* $X \equiv \lambda Y.(X\sqsubseteq Y)$

Postulate: $\mathcal{HF}\ \mathcal{G}$ is a filter; i.e., Hauptfilter of $\mathcal{G}$.

axiomatization where *F1*: $\lfloor$*Filter* (*HF* $\mathcal{G}$)$\rfloor$

Necessary existence of a Godlike entity.

theorem *T6*: $\lfloor\Box(\exists^E\ \mathcal{G})\rfloor$ **using** *F1* **by** *auto* — Proof found

theorem *T6again*: $\lfloor\Box(\exists^E\ \mathcal{G})\rfloor$
proof –
have *T3'*: $\lfloor\exists^E\ \mathcal{G}\rfloor$ **using** *F1* **by** *auto*
have *T6*: $\lfloor\Box(\exists^E\ \mathcal{G})\rfloor$ **using** *T3'* **by** *blast*

thus *?thesis* **by** *simp* **qed**

Possible existence of Godlike entity not implied.

lemma *T3*: $\lfloor\Diamond(\exists^E\ \mathcal{G})\rfloor$ **nitpick oops** — Countermodel

Axiom T enforces possible existence of Godlike entity.

axiomatization
lemma *T3*: **assumes** *T*: $\lfloor\forall\,\varphi.((\Box\varphi) \rightarrow \varphi)\rfloor$
shows $\lfloor\Diamond(\exists^E\ \mathcal{G})\rfloor$ **using** *F1 T* **by** *auto*

lemma *True* **nitpick**[*satisfy*] **oops** — Consistency: model found

Modal collapse: not implied anymore.

lemma *MC*: $\lfloor\forall\,\Phi.(\Phi \rightarrow \Box\Phi)\rfloor$ **nitpick oops** — Countermodel
lemma *MT*: $\lfloor\forall\,x\ y.(((\mathcal{G}\ x) \wedge (\mathcal{G}\ y)) \rightarrow (x{=}y))\rfloor$
nitpick oops — Countermodel
end

4.10 Formal Study of Version No.2 of Gödel's Argument as Reported by Kanckos and Lethen, 2019 [6] (Fig. 11 in [4])

theory *KanckosLethenNo2Possibilist* **imports**
HOML
MFilter
BaseDefs
begin

Axioms of Version No. 2 [6].

abbreviation *delta* (Δ) **where** $\Delta\ A \equiv \lambda x.(\forall\,\psi.\ ((A\ \psi) \rightarrow (\psi\ x)))$
abbreviation *N* $(\mathcal{N})$ **where** $\mathcal{N}\ \varphi \equiv \lambda x.(\Box(\varphi\ x))$

axiomatization where
Axiom1: $\lfloor\forall\,\varphi\ \psi.(((\mathcal{P}\ \varphi) \wedge (\Box(\forall\,x.\ ((\varphi\ x) \rightarrow (\psi\ x))))) \rightarrow (\mathcal{P}\ \psi))\rfloor$ **and** — The $\Box$ can be omitted here; the proofs still work.
Axiom2: $\lfloor\forall\,A\ .(\Box((\forall\,\varphi.((A\ \varphi) \rightarrow\ (\mathcal{P}\ \varphi))) \rightarrow (\mathcal{P}\ (\Delta\ A))))\rfloor$ **and** — The $\Box$ can be omitted here; the proofs still work.
Axiom3: $\lfloor\forall\,\varphi.((\mathcal{P}\ \varphi) \rightarrow (\mathcal{P}\ (\mathcal{N}\ \varphi)))\rfloor$ **and**
Axiom4: $\lfloor\forall\,\varphi.((\mathcal{P}\ \varphi) \rightarrow (\neg(\mathcal{P}(\rightharpoondown\varphi))))\rfloor$ **and**
— Logic S5
axB: $\lfloor\forall\,\varphi.(\varphi \rightarrow \Box\Diamond\varphi)\rfloor$ **and** *axM*: $\lfloor\forall\,\varphi.((\Box\varphi) \rightarrow \varphi)\rfloor$ **and** *ax4*: $\lfloor\forall\,\varphi.((\Box\varphi) \rightarrow (\Box\Box\varphi))\rfloor$

Sahlqvist correspondences: they are better suited for proof automation.

lemma *axB′*: $\forall\,x\ y.\ \neg(x\mathbf{r}y) \vee (y\mathbf{r}x)$ **using** *axB* **by** *fastforce*
lemma *axM′*: $\forall\,x.\ (x\mathbf{r}x)$ **using** *axM* **by** *blast*
lemma *ax4′*: $\forall\,x\ y\ z.\ (((x\mathbf{r}y) \wedge (y\mathbf{r}z)) \longrightarrow (x\mathbf{r}z))$ **using** *ax4* **by** *auto*

Proofs for all theorems for No.2 from [6].

theorem *Theorem0*: $\lfloor\forall\varphi\ \psi.((\forall Q.\ ((Q\ \varphi)\ \rightarrow (Q\ \psi))) \rightarrow\ ((\mathcal{P}\ \varphi) \rightarrow (\mathcal{P}\ \ \varphi)))\rfloor$ **by** *auto* — not needed
theorem *Theorem1*: $\lfloor\mathcal{P}\ \mathcal{G}\rfloor$ **unfolding** *G-def* **using** *Axiom2 axM* **by** *blast*
theorem *Theorem2*: $\lfloor\forall x.\ ((\mathcal{G}\ x)\rightarrow(\exists y.\ \mathcal{G}\ y))\rfloor$ **by** *blast* — not needed
theorem *Theorem3a*: $\lfloor\mathcal{P}\ (\lambda x.\ (\exists y.\ \mathcal{G}\ y))\rfloor$ **by** (*metis* (*no-types*, *lifting*) *Axiom1 Theorem1*)
theorem *Theorem3b*: $\lfloor\Box(\mathcal{P}\ (\lambda x.(\Box(\exists y.\ \mathcal{G}\ y))))\rfloor$ **by** (*smt Axiom1 G-def Theorem3a Axiom3 Theorem1 axB'*)
theorem *Theorem4*: $\lfloor\forall x.\ \Box((\mathcal{G}\ x) \rightarrow ((\mathcal{P}\ (\lambda x.(\Box(\exists y.\ \mathcal{G}\ y)))) \rightarrow\ (\Box(\exists y.\ \mathcal{G}\ y))))\rfloor$ **using** *G-def* **by** *fastforce* — not needed
theorem *Theorem5*: $\lfloor\forall x.\ \Box((\mathcal{G}\ x)\ \rightarrow\ (\Box(\exists y.\ \mathcal{G}\ y)))\rfloor$ **by** (*smt* (*verit*) *G-def Theorem3a Theorem3b*) — not needed
theorem *Theorem6*: $\lfloor\Box((\exists y.\ \mathcal{G}\ y) \rightarrow (\Box(\exists y.\ \mathcal{G}\ y)))\rfloor$ **by** (*smt G-def Theorem3a Theorem3b*)
theorem *Theorem7*: $\lfloor\Box((\Diamond(\exists y.\ \mathcal{G}\ y)) \rightarrow (\Box(\exists y.\ \mathcal{G}\ y)))\rfloor$ **using** *Theorem6 axB'* **by** *blast*
theorem *Theorem8*: $\lfloor\Box(\exists y.\ \mathcal{G}\ y)\rfloor$ **by** (*metis Axiom1 Axiom4 Theorem1 Theorem7 axB'*)
theorem *Theorem9*: $\lfloor\forall\varphi.\ ((\mathcal{P}\ \varphi) \rightarrow \Diamond(\exists x.\ \varphi\ x))\rfloor$ **using** *Axiom1 Axiom4 axM'* **by** *metis*

Short proof of Theorem8; analogous to the one presented in Sec. 7 of Benzmüller 2020.

theorem $\lfloor\Box(\exists y.\ \mathcal{G}\ y)\rfloor$ — Note: this version of the proof uses only *axB'* and *axM'*.
proof –
have *L1*: $\lfloor(\exists X.((\mathcal{P}\ X)\wedge\neg(\exists X)))\rightarrow(\mathcal{P}(\lambda x.(x\neq x)))\rfloor$ **using** *Axiom1 Axiom3 axB'* **by** *blast* — Use metis here if □ is omitted in Axiom1 and Axiom 2
have *L2*: $\lfloor\neg(\mathcal{P}(\lambda x.(x\neq x)))\rfloor$ **using** *Axiom1 Axiom4* **by** *metis*
have *L3*: $\lfloor\neg(\exists X.((\mathcal{P}\ X)\ \wedge\ \neg(\exists\ \ X)))\rfloor$ **using** *L1 L2* **by** *blast*
have *T2*: $\lfloor\mathcal{P}\ \mathcal{G}\rfloor$ **by** (*smt Axiom1 Axiom2 G-def axM'*)
have *T3*: $\lfloor\exists y.\ \mathcal{G}\ y\rfloor$ **using** *L3 T2* **by** *blast*
have *T6*: $\lfloor\Box(\exists y.\ \mathcal{G}\ y)\rfloor$ **by** (*simp add*: *T3*)
thus *?thesis* **by** *blast* **qed**

theorem *T5*: $\lfloor(\Diamond(\exists y.\ \mathcal{G}\ y)) \rightarrow \Box(\exists y.\ \mathcal{G}\ y)\rfloor$ — Obvious: If we can prove Theorem8, then we also have T5.
proof –
have *L1*: $\lfloor(\exists X.((\mathcal{P}\ X)\wedge\neg(\exists X)))\rightarrow(\mathcal{P}(\lambda x.(x\neq x)))\rfloor$ **using** *Axiom1 Axiom3 axB'* **by** *blast* — Use metis here if □ is omitted in Axiom1 and Axiom 2
have *L2*: $\lfloor\neg(\mathcal{P}(\lambda x.(x\neq x)))\rfloor$ **using** *Axiom1 Axiom4* **by** *metis*
have *L3*: $\lfloor\neg(\exists X.((\mathcal{P}\ X)\ \wedge\ \neg(\exists\ \ X)))\rfloor$ **using** *L1 L2* **by** *blast*
have *T2*: $\lfloor\mathcal{P}\ \mathcal{G}\rfloor$ **by** (*smt Axiom1 Axiom2 G-def axM'*)
have *T3*: $\lfloor\exists y.\ \mathcal{G}\ y\rfloor$ **using** *L3 T2* **by** *blast*
have *T6*: $\lfloor\Box(\exists y.\ \mathcal{G}\ y)\rfloor$ **by** (*simp add*: *T3*)
thus *?thesis* **by** *blast* **qed**

Another short proof of Theorem8.

theorem $\lfloor\Box(\exists y.\ \mathcal{G}\ y)\rfloor$ — Note: fewer assumptions used in some cases than in [6].

proof –
have *T1*: $\lfloor \mathcal{P}\ \mathcal{G} \rfloor$ **unfolding** *G-def* **using** *Axiom2 axM* **by** *blast*
have *T3a*: $\lfloor \mathcal{P}\ (\lambda x.\ (\exists\, y.\ \mathcal{G}\ y)) \rfloor$ **by** (*metis* (*no-types*, *lifting*) *Axiom1 T1*)
have *T3b*: $\lfloor \Box(\mathcal{P}\ (\lambda x.(\Box(\exists\, y.\ \mathcal{G}\ y)))) \rfloor$ **by** (*smt Axiom1 G-def T3a Axiom3 T1 axB′*)
have *T6*: $\lfloor \Box((\exists\, y.\ \mathcal{G}\ y) \rightarrow (\Box(\exists\, y.\ \mathcal{G}\ y))) \rfloor$ **by** (*smt G-def T3a T3b*)
have *T7*: $\lfloor \Box((\Diamond(\exists\, y.\ \mathcal{G}\ y)) \rightarrow (\Box(\exists\, y.\ \mathcal{G}\ y))) \rfloor$ **using** *T6 axB′* **by** *blast*
thus *?thesis* **by** (*smt Axiom1 Axiom4 T3b axB′*) **qed**

Are the axioms of the simplified versions implied?

Actualist version of the axioms.

lemma *A1′*: $\lfloor \neg(\mathcal{P}(\lambda x.(x{\neq}x))) \rfloor$ **using** *Theorem9* **by** *blast*
lemma *A2′*: $\lfloor \forall\, X\ Y.(((\mathcal{P}\ X) \wedge ((X \sqsubseteq Y) \vee (X \Rrightarrow Y))) \rightarrow (\mathcal{P}\ Y)) \rfloor$ **nitpick oops** — Countermodel
lemma *A3*: $\lfloor \forall\, \mathcal{Z}.((\mathcal{P}\mathrm{os}\ \mathcal{Z}) \rightarrow (\forall\, X.((X \bigsqcap \mathcal{Z}) \rightarrow (\mathcal{P}\ X)))) \rfloor$ **nitpick oops** — Countermodel

Possibilist version of the axioms.

abbreviation *a* (-$\sqsubseteq^p$-) **where** $X \sqsubseteq^p Y \equiv \forall\, z.((X\ z) \rightarrow (Y\ z))$
abbreviation *b* (-$\Rrightarrow^p$-) **where** $X \Rrightarrow^p Y \equiv \Box(X \sqsubseteq^p Y)$
abbreviation *d* (-$\bigsqcap^p$-) **where** $X \bigsqcap^p \mathcal{Z} \equiv \Box(\forall\, u.((X\ u) \leftrightarrow (\forall\, Y.((\mathcal{Z}\ Y) \rightarrow (Y\ u)))))$

lemma *A1′P*: $\lfloor \neg(\mathcal{P}(\lambda x.(x{\neq}x))) \rfloor$ **using** *Theorem9* **by** *blast*
lemma *A2′P*: $\lfloor \forall\, X\ Y.(((\mathcal{P}\ X) \wedge ((X \sqsubseteq^p Y) \vee (X \Rrightarrow^p Y))) \rightarrow (\mathcal{P}\ Y)) \rfloor$ **oops** — no answer, yet by sledgehammer and nitpick
lemma *A2′aP*: $\lfloor \forall\, X\ Y.(((\mathcal{P}\ X) \wedge (X \Rrightarrow^p Y)) \rightarrow (\mathcal{P}\ Y)) \rfloor$ **using** *Axiom1 axM′* **by** *metis*
lemma *A2′bP*: $\lfloor \forall\, X\ Y.(((\mathcal{P}\ X) \wedge (X \sqsubseteq^p Y)) \rightarrow (\mathcal{P}\ Y)) \rfloor$ **oops** — no answer, yet by sledgehammer and nitpick
lemma *A3P*: $\lfloor \forall\, \mathcal{Z}.((\mathcal{P}\mathrm{os}\ \mathcal{Z}) \rightarrow (\forall\, X.((X \bigsqcap^p \mathcal{Z}) \rightarrow (\mathcal{P}\ X)))) \rfloor$
by (*smt* (*verit*, *del-insts*) *Axiom1 Axiom2 axM′*) — proof found

Are Axiom2 and A3 equivalent? Only when assuming Axiom1 and axiom M.

lemma $\lfloor \forall\, A\ .(\Box((\forall\, \varphi.((A\ \varphi) \rightarrow (\mathcal{P}\ \varphi))) \rightarrow (\mathcal{P}\ (\Delta\ A)))) \rfloor \equiv \lfloor \forall\, \mathcal{Z}.((\mathcal{P}\mathrm{os}\ \mathcal{Z}) \rightarrow (\forall\, X.((X \bigsqcap^p \mathcal{Z}) \rightarrow (\mathcal{P}\ X)))) \rfloor$
by (*smt* (*verit*, *ccfv-threshold*) *Axiom1 axM′*) — proof found
end

References

[1] C. Benzmüller and D. Fuenmayor. Computer-supported analysis of positive properties, ultrafilters and modal collapse in variants of Gödel's ontological argument. *Bulletin of the Section of Logic*, 49(2):127–148, 2020.

[2] C. Benzmüller and L. Paulson. Quantified multimodal logics in simple type theory. *Logica Universalis*, 7(1):7–20, 2013.

[3] C. Benzmüller and B. Woltzenlogel Paleo. Gödel's God in Isabelle/HOL. *Archive of Formal Proofs*, 2013, 2013.

[4] C. Benzmüller. A (Simplified) Supreme Being Necessarily Exists, says the Computer: Computationally Explored Variants of Gödel's Ontological Argument. In *Proceedings of the 17th International Conference on Principles of Knowledge Representation and Reasoning, KR 2020*, pages 779–789. IJCAI organization, 9 2020.

[5] K. Gödel. Appendix A. Notes in Kurt Gödel's Hand. In Sobel [8], pages 144–145.

[6] A. Kanckos and T. Lethen. The development of Gödel's ontological proof. *The Review of Symbolic Logic*, 11 2019.

[7] D. S. Scott. Appendix B: Notes in Dana Scott's Hand. In Sobel [8], pages 145–146.

[8] J. H. Sobel. *Logic and Theism: Arguments for and Against Beliefs in God*. Cambridge University Press, 2004.